Right division

```
In[•]:= L1 = (x^2 - 1) y'''[x] - 2 x y''[x] + x^2 y'[x] - y[x];
        L2 = (x + 1) y''[x] + (2 - x) y[x];
        Qot = 0;
```

```
In[•]:= L1a = L1 - (x^2 - 1) / (x + 1) D[L2, x] // Simplify
        Qot += Factor[(x^2 - 1 ) / (x + 1)] ** Dx
```

Out[•]=

$(-2 + x)\, y[x] + \left(2 - 3\, x + 2\, x^2\right) y'[x] + \left(1 - 3\, x\right) y''[x]$

Out[•]=

$(-1 + x) ** \mathrm{Dx}$

```
In[•]:= Rem = L1a - (1 - 3 x) / (x + 1) L2 // Simplify
        Qot += (1 - 3 x) / (x + 1)
```

Out[•]=

$$\frac{-2\left(2 - 3\, x + x^2\right) y[x] + \left(2 - x - x^2 + 2\, x^3\right) y'[x]}{1 + x}$$

Out[•]=

$$\frac{1 - 3\, x}{1 + x} + (-1 + x) ** \mathrm{Dx}$$

Least common left multiple

```
In[•]:= L1 = y'[x] - x y[x]
        L2 = y''[x] - 2 x  y'[x] + (x + 1) y[x]
```

Out[•]=

$-x\, y[x] + y'[x]$

Out[•]=

$(1 + x)\, y[x] - 2\, x\, y'[x] + y''[x]$

```
In[•]:= L = Sum[l[i] × D[y[x], {x, i}], {i, 0, 3}]
```

Out[•]=

$l[0] \times y[x] + l[1]\, y'[x] + l[2]\, y''[x] + l[3]\, y^{(3)}[x]$

```
In[•]:= L1a = L - Coefficient[L, Derivative[3][y][x]] × D[L1, {x, 2}] // Simplify
```

Out[•]=

$l[0] \times y[x] + \left(l[1] + 2\, l[3]\right) y'[x] + \left(l[2] + x\, l[3]\right) y''[x]$

```
In[•]:= L1b = L1a - Coefficient[L1a, Derivative[2][y][x]] × D[L1, x] // Simplify
```

Out[•]=

$\left(l[0] + l[2] + x\, l[3]\right) y[x] + \left(l[1] + x\, l[2] + 2\, l[3] + x^2\, l[3]\right) y'[x]$

In[ ]:=
```
lremLL1 = L1b - Coefficient[L1b, y'[x]] L1 // Simplify
```

Out[ ]=

$$\left(l[0]+l[2]+x^2\,l[2]+x^3\,l[3]+x\,\left(l[1]+3\,l[3]\right)\right)\,y[x]$$

In[ ]:=
```
L2a = L - Coefficient[L, y'''[x]] × D[L2, x] // Simplify
```

Out[ ]=

$$(l[0]-l[3])\,y[x]+\left(l[1]+l[3]-x\,l[3]\right)\,y'[x]+\left(l[2]+2\,x\,l[3]\right)\,y''[x]$$

In[ ]:=
```
lremLL2 = L2a - Coefficient[L2a, y''[x]] L2 // Simplify
```

Out[ ]=

$$\left(l[0]-(1+x)\,l[2]-\left(1+2\,x+2\,x^2\right)\,l[3]\right)\,y[x]+\left(l[1]+2\,x\,l[2]+l[3]-x\,l[3]+4\,x^2\,l[3]\right)\,y'[x]$$

In[ ]:=
```
sys = {Coefficient[lremLL1, y[x]], Coefficient[lremLL2, y[x]], Coefficient[lremLL2, y'[x]]}
```

Out[ ]=

$$\left\{l[0]+l[2]+x^2\,l[2]+x^3\,l[3]+x\,\left(l[1]+3\,l[3]\right),\; l[0]-(1+x)\,l[2]-\left(1+2\,x+2\,x^2\right)\,l[3],\; l[1]+2\,x\,l[2]+l[3]-x\,l[3]+4\,x^2\,l[3]\right\}$$

In[ ]:=
```
sol = Solve[sys == 0, {l[0], l[1], l[2]}]
```

Out[ ]=

$$\left\{\left\{l[0]\to-\frac{(1+x)\,\left(l[3]-2\,x\,l[3]+x^2\,l[3]\right)}{-2+x},\; l[1]\to-\frac{-2\,l[3]+3\,x\,l[3]+2\,x^2\,l[3]+x^3\,l[3]-2\,x^4\,l[3]}{(-2+x)\,(1+x)},\; l[2]\to-\frac{-l[3]-4\,x\,l[3]-3\,x^2\,l[3]+3\,x^3\,l[3]}{(-2+x)\,(1+x)}\right\}\right\}$$

Choose l[3] to be polynomial

In[ ]:=
```
sol = First[sol] /. l[3] → (x + 1) (x - 2) // Factor
```

Out[ ]=

$$\left\{l[0]\to-(-1+x)^2\,(1+x)^2,\; l[1]\to\left(-1+2\,x\right)\left(-2-x+x^3\right),\; l[2]\to1+4\,x+3\,x^2-3\,x^3\right\}$$

In[ ]:=
```
LCLM = L /. sol /. l[3] → (x + 1) (x - 2)
```

Out[ ]=

$$-(-1+x)^2\,(1+x)^2\,y[x]+\left(-1+2\,x\right)\left(-2-x+x^3\right)\,y'[x]+\left(1+4\,x+3\,x^2-3\,x^3\right)\,y''[x]+(-2+x)\,(1+x)\,y^{(3)}[x]$$

In[ ]:=
```
DSolve[L1 == 0, y[x], x][[1, 1, 2]]
DSolve[L2 == 0, y[x], x][[1, 1, 2]]
```

Out[ ]=

$$e^{\frac{x^2}{2}}\,c_1$$

Out[ ]=

$$e^{x/2}\,c_1\,\text{HermiteH}\left[\frac{5}{8},\,-\frac{1}{2}+x\right]+e^{x/2}\,c_2\,\text{Hypergeometric1F1}\left[-\frac{5}{16},\,\frac{1}{2},\,\left(-\frac{1}{2}+x\right)^2\right]$$

```
In[∘]:= f[1, x_] := e^(x^2/2) c0;
        f[2, x_] := e^(x/2) c1 HermiteH[5/8, -1/2 + x] + e^(x/2) c2 Hypergeometric1F1[-5/16, 1/2, (-1/2 + x)^2];
```

```
In[∘]:= f[x_] := e^(x^2/2) c0 + e^(x/2) c1 HermiteH[5/8, -1/2 + x] + e^(x/2) c2 Hypergeometric1F1[-5/16, 1/2, (-1/2 + x)^2]
```

```
In[∘]:= LCLM /. y → f // FullSimplify
Out[∘]=
0
```

Closure property “times” for holonomic functions:

```
In[∘]:= ode1 = f1''[x] == (x + 1) f1'[x] + f1[x];
        ode2 = f2''[x] == 2 x f2'[x] - x^2 f2[x];
```

```
In[∘]:= DSolve[ode1, f1[x], x]
        DSolve[ode2, f2[x], x]
```

Out[∘]=

$$\left\{\left\{f1[x] \to e^{x+\frac{x^2}{2}}\, c_2 + e^{\frac{1}{2}+x+\frac{x^2}{2}} \sqrt{\frac{\pi}{2}}\, c_1\, \mathrm{Erf}\left[\frac{1+x}{\sqrt{2}}\right]\right\}\right\}$$

Out[∘]=

$$\left\{\left\{f2[x] \to e^{\frac{1}{2} x (-2 i+x)}\, c_1 - \frac{1}{2}\, i\, e^{2 i x+\frac{1}{2} x (-2 i + x)}\, c_2\right\}\right\}$$

Express higher order derivatives in terms of f1, f2

```
In[∘]:= subst12 = f1''[x] → (x + 1) f1'[x] + f1[x] ;
        subst13 = f1'''[x] → Collect[D[ode1[[2]], x] /. subst12, Derivative[_.][f1][x], Factor];
        subst14 = f1''''[x] →
           Collect[D[ode1[[2]], {x, 2}] /. subst13 /. subst12, Derivative[_.][f1][x], Factor];
        subst22 = f2''[x] → 2 x f2'[x] - x^2 f2[x];
        subst23 = f2'''[x] → Collect[D[ode2[[2]], x] /. subst22, Derivative[_.][f2][x], Factor];
        subst24 = f2''''[x] →
           Collect[D[ode2[[2]], {x, 2}] /. subst23 /. subst22, Derivative[_.][f2][x], Factor];
```

```
In[∘]:= subst13
```

Out[∘]=

$$f1^{(3)}[x] \to (1+x)\, f1[x] + \left(3 + 2x + x^2\right) f1'[x]$$

```
In[∘]:= subst24
```

Out[∘]=

$$f2^{(4)}[x] \to -\left(\left(2 + 8x^2 + 3x^4\right) f2[x]\right) + 4x\left(2 + x^2\right) f2'[x]$$

Ansatz for the order 4 differential equation of the product f = f1 f2

```
In[∘]:= f[x_] := f1[x] × f2[x]
```

In[ ]:=
```
ans = Sum[a[i] × D[f[x], {x, i}], {i, 0, 4}] /.
  {subst14, subst24, subst13, subst23, subst12, subst22}
```

Out[ ]=

$$a[0]\, f1[x]\, f2[x] + a[1] \left(f2[x]\, f1'[x] + f1[x]\, f2'[x]\right) + a[2] \left(f2[x] \left(f1[x] + (1+x)\, f1'[x]\right) + 2\, f1'[x]\, f2'[x] + f1[x] \left(-x^2\, f2[x] + 2\, x\, f2'[x]\right)\right) + a[3] \left(f2[x] \left((1+x)\, f1[x] + \left(3 + 2\, x + x^2\right) f1'[x]\right) + 3 \left(f1[x] + (1+x)\, f1'[x]\right) f2'[x] + 3\, f1'[x] \left(-x^2\, f2[x] + 2\, x\, f2'[x]\right) + f1[x] \left(-2\, x \left(1 + x^2\right) f2[x] + \left(2 + 3\, x^2\right) f2'[x]\right)\right) + a[4] \left(f2[x] \left(\left(4 + 2\, x + x^2\right) f1[x] + (1+x) \left(6 + 2\, x + x^2\right) f1'[x]\right) + 4 \left((1+x)\, f1[x] + \left(3 + 2\, x + x^2\right) f1'[x]\right) f2'[x] + 6 \left(f1[x] + (1+x)\, f1'[x]\right) \left(-x^2\, f2[x] + 2\, x\, f2'[x]\right) + f1[x] \left(-\left(\left(2 + 8\, x^2 + 3\, x^4\right) f2[x]\right) + 4\, x \left(2 + x^2\right) f2'[x]\right) + 4\, f1'[x] \left(-2\, x \left(1 + x^2\right) f2[x] + \left(2 + 3\, x^2\right) f2'[x]\right)\right)$$

Extract coefficients of the linear independent terms

In[ ]:=
```
cl = Coefficient[ans, {f1[x] × f2[x], f1'[x] × f2[x], f1[x] × f2'[x], f1'[x] × f2'[x]}] // Factor
```

Out[ ]=

$$\{a[0] + a[2] - x^2\, a[2] + a[3] - x\, a[3] - 2\, x^3\, a[3] + 2\, a[4] + 2\, x\, a[4] - 13\, x^2\, a[4] - 3\, x^4\, a[4],\ a[1] + a[2] + x\, a[2] + 3\, a[3] + 2\, x\, a[3] - 2\, x^2\, a[3] + 6\, a[4] - 3\, x^2\, a[4] - 13\, x^3\, a[4],\ a[1] + 2\, x\, a[2] + 5\, a[3] + 3\, x^2\, a[3] + 4\, a[4] + 24\, x\, a[4] + 4\, x^3\, a[4],\ 2\, a[2] + 3\, a[3] + 9\, x\, a[3] + 20\, a[4] + 20\, x\, a[4] + 28\, x^2\, a[4]\}$$

Solve for the coefficients of the order 4 DE

In[ ]:=
```
sol = Solve[Thread[cl == 0], {a[0], a[1], a[2], a[3], a[4]}] // Factor
```

Solve: Equations may not give solutions for all "solve" variables.

Out[ ]=

$$\left\{\left\{a[0] \to \frac{4\, (1+x)\, (2+x) \left(6 - 14\, x + 5\, x^2 + x^4\right) a[2]}{-46 + 54\, x + 109\, x^2 + 34\, x^3 + 13\, x^4},\ a[1] \to -\frac{2 \left(-26 - 78\, x + 21\, x^2 + 48\, x^3 + 17\, x^4 + 6\, x^5\right) a[2]}{-46 + 54\, x + 109\, x^2 + 34\, x^3 + 13\, x^4},\ a[3] \to -\frac{2 \left(8 + 24\, x + 7\, x^2 + 3\, x^3\right) a[2]}{-46 + 54\, x + 109\, x^2 + 34\, x^3 + 13\, x^4},\ a[4] \to \frac{\left(7 + 2\, x + x^2\right) a[2]}{-46 + 54\, x + 109\, x^2 + 34\, x^3 + 13\, x^4}\right\}\right\}$$

Choose a[2] to have a polynomial equation

In[ ]:=
```
sol = sol /. a[2] → -46 + 54 x + 109 x^2 + 34 x^3 + 13 x^4
```

Out[ ]=

$$\left\{\left\{a[0] \to 4\, (1+x)\, (2+x) \left(6 - 14\, x + 5\, x^2 + x^4\right),\ a[1] \to -2 \left(-26 - 78\, x + 21\, x^2 + 48\, x^3 + 17\, x^4 + 6\, x^5\right),\ a[3] \to -2 \left(8 + 24\, x + 7\, x^2 + 3\, x^3\right),\ a[4] \to 7 + 2\, x + x^2\right\}\right\}$$

Plug in to obtain the DE for f=f1 f2

In[ ]:=
```
ann0 = Sum[a[i] × D[y[x], {x, i}], {i, 0, 4}] /. sol[[1]] /. a[2] → -46 + 54 x + 109 x^2 + 34 x^3 + 13 x^4
```

Out[ ]=

$$4 (1 + x) (2 + x) \left(6 - 14 x + 5 x^2 + x^4\right) y[x] - 2 \left(-26 - 78 x + 21 x^2 + 48 x^3 + 17 x^4 + 6 x^5\right) y'[x] + \left(-46 + 54 x + 109 x^2 + 34 x^3 + 13 x^4\right) y''[x] - 2 \left(8 + 24 x + 7 x^2 + 3 x^3\right) y^{(3)}[x] + \left(7 + 2 x + x^2\right) y^{(4)}[x]$$

HolonomicFunctions by Christoph Koutschan available at
https://risc.jku.at/sw/holonomicfunctions/
via
http://koutschan.de/software.php
or (as part of the RISC ErgoSum bundle)
https://combinatorics.risc.jku.at/software

In[ ]:=
```
<< RISC`HolonomicFunctions`
```

HolonomicFunctions Package version 1.7.3 (21-Mar-2017)
written by Christoph Koutschan
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

```
--> Type  ?HolonomicFunctions  for help.
```

In[ ]:=
```
? HolonomicFunctions
```

Out[ ]=

Symbol

The main objective of this package is the algorithmic manipulation of ∂–finite (holonomic) functions. This includes (but is not restricted to) proving special function identities, finding recurrences, differential equations or relations of mixed type for ∂–finite functions, and computing definite sums and integrals of ∂–finite functions. Type ?DFinite to get the definition and a short introduction to ∂–finite functions.

The following commands serve the above objectives: Annihilator, CreativeTelescoping, HermiteTelescoping, FindCreativeTelescoping, FindRelation, FindSupport, Takayama, ApplyOreOperator, UnderTheStaircase, AnnihilatorDimension.

The closure properties of ∂–finite functions are implicitely executed in Annihilator. To execute them explicitly, use the commands DFinitePlus, DFiniteTimes, DFiniteSubstitute, DFiniteOreAction, DFiniteTimesHyper, DFiniteDE2RE, DFiniteRE2DE, DFiniteQSubstitute.

An important ingredient are Groebner bases in (noncommutative) Ore algebras: OreGroebnerBasis, OreReduce, GBEqual, FGLM.

A common subtask in the above algorithms is finding rational solutions of P–finite recurrences / differential equations or of coupled systems of such equations. The following commands address these purposes: RSolvePolynomial, RSolveRational, DSolvePolynomial, DSolveRational, QSolvePolynomial, QSolveRational, SolveOreSys, SolveCoupledSystem.

An element of an Ore algebra is called an Ore polynomial; the following commands explain the data type OrePolynomial that is introduced in this package and how to deal with it: OrePolynomial, ToOrePolynomial, OrePolynomialZeroQ, LeadingPowerProduct, LeadingExponent, LeadingCoefficient, LeadingTerm, OrePolynomialListCoefficients, NormalizeCoefficients, OrePlus, OreTimes, OrePower, ApplyOreOperator, ChangeOreAlgebra, ChangeMonomialOrder, OrePolynomialSubstitute, OrePolynomialDegree, Support.

In order to define own Ore algebras use the commands OreAlgebra, OreAlgebraGenerators, OreAlgebraOperators, OreAlgebraPolynomialVariables, OreOperators, OreOperatorQ, OreSigma, OreDelta, OreAction, Der, S, Delta, Euler, QS.

Some other functions that might be useful: Printlevel, RandomPolynomial.

If this package was useful in your scientific work, proper citation would be appreciated very much. Please use the following reference for this purpose:

```
@phdthesis{Koutschan09,
  author = {Christoph Koutschan},
  title = {Advanced Applications of the Holonomic Systems Approach},
  school = {RISC, Johannes Kepler University},
  address = {Linz, Austria},
  year = {2009}
}
```

In[ ]:= **ode1**

Out[ ]=

$f1''[x] == f1[x] + (1 + x)\, f1'[x]$

In[ ]:= **ann1 = ToOrePolynomial[ode1[[1]] - ode1[[2]], f1[x]]**

Out[ ]=

$D_x^2 + (-1 - x)\, D_x - 1$

In[ ]:= **ann2 = ToOrePolynomial[ode2[[1]] - ode2[[2]], f2[x]]**

Out[ ]=

$D_x^2 - 2\, x\, D_x + x^2$

In[ ]:= `ann = DFiniteTimes[{ann1}, {ann2}] // Factor`

Out[ ]=

$$\left\{\left(7+2x+x^2\right)D_x^4-2\left(8+24x+7x^2+3x^3\right)D_x^3+\left(-46+54x+109x^2+34x^3+13x^4\right)D_x^2-2\left(-26-78x+21x^2+48x^3+17x^4+6x^5\right)D_x+4(1+x)(2+x)\left(6-14x+5x^2+x^4\right)\right\}$$

In[ ]:= `ann0`

Out[ ]=

$$4(1+x)(2+x)\left(6-14x+5x^2+x^4\right)y[x]-2\left(-26-78x+21x^2+48x^3+17x^4+6x^5\right)y'[x]+\left(-46+54x+109x^2+34x^3+13x^4\right)y''[x]-2\left(8+24x+7x^2+3x^3\right)y^{(3)}[x]+\left(7+2x+x^2\right)y^{(4)}[x]$$

In[ ]:= `ToOrePolynomial[ann0, y[x]]`

Out[ ]=

$$\left(7+2x+x^2\right)D_x^4+\left(-16-48x-14x^2-6x^3\right)D_x^3+\left(-46+54x+109x^2+34x^3+13x^4\right)D_x^2+\left(52+156x-42x^2-96x^3-34x^4-12x^5\right)D_x+\left(48-40x-104x^2+4x^3+28x^4+12x^5+4x^6\right)$$

In[ ]:= `Factor[%]`

Out[ ]=

$$\left(7+2x+x^2\right)D_x^4-2\left(8+24x+7x^2+3x^3\right)D_x^3+\left(-46+54x+109x^2+34x^3+13x^4\right)D_x^2-2\left(-26-78x+21x^2+48x^3+17x^4+6x^5\right)D_x+4(1+x)(2+x)\left(6-14x+5x^2+x^4\right)$$

In[ ]:= `f[1, x]`

Out[ ]=

$$e^{\frac{x^2}{2}}\,\mathbb{c}_0$$

In[ ]:= `annF1 = Annihilator[f[1, x], Der[x]]`

Out[ ]=

$$\{D_x-x\}$$

In[ ]:= `annF2 = Annihilator[f[2, x], Der[x]]`

Out[ ]=

$$\left\{D_x^2-2x\,D_x+(1+x)\right\}$$

this time DFinitePlus worked just fine (no changes or update needed)

In[ ]:= `DFinitePlus[annF1, annF2]`

Out[ ]=

$$\left\{\left(-2-x+x^2\right)D_x^3+\left(1+4x+3x^2-3x^3\right)D_x^2+\left(2-3x-2x^2-x^3+2x^4\right)D_x+\left(-1+2x^2-x^4\right)\right\}$$

In[ ]:= `Annihilator[f[1, x] + f[2, x], Der[x]] // Factor`

Out[ ]=

$$\left\{(-2+x)(1+x)D_x^3+\left(1+4x+3x^2-3x^3\right)D_x^2+\left(-1+2x\right)\left(-2-x+x^3\right)D_x-(-1+x)^2(1+x)^2\right\}$$

In[ ]:= `LCLM`

Out[ ]=

$$-(-1+x)^2(1+x)^2y[x]+\left(-1+2x\right)\left(-2-x+x^3\right)y'[x]+\left(1+4x+3x^2-3x^3\right)y''[x]+(-2+x)(1+x)y^{(3)}[x]$$

In[ ]:=
```
annE = Annihilator[Exp[x y], {Der[x], Der[y]}]
```

Out[ ]=

$$\{D_y - x,\ D_x - y\}$$

In[ ]:=
```
annS = Annihilator[Sqrt[x y], {Der[x], Der[y]}]
```

Out[ ]=

$$\{2\, y\, D_y - 1,\ 2\, x\, D_x - 1\}$$

In[ ]:=
```
? DFinite*
```

Out[ ]=

RISC`HolonomicFunctions`

DFinite	DFiniteOreAction	DFiniteQSubstitute	DFiniteSubstitute	DFiniteTimesHyper
DFiniteDE2RE	DFinitePlus	DFiniteRE2DE	DFiniteTimes	

In[ ]:=
```
ann = DFinitePlus[annE, annS]
```

Out[ ]=

$$\{x\, D_x - y\, D_y,\ (-2\, y + 4\, x\, y^2)\, D_y^2 + (-1 - 4\, x^2\, y^2)\, D_y + (x + 2\, x^2\, y)\}$$

In[ ]:=
```
annE = Annihilator[Exp[x y], {Der[y], Der[x]}]
annS = Annihilator[Sqrt[x y], {Der[y], Der[x]}]
```

Out[ ]=

$$\{D_x - y,\ D_y - x\}$$

Out[ ]=

$$\{2\, x\, D_x - 1,\ 2\, y\, D_y - 1\}$$

In[ ]:=
```
ann = DFinitePlus[annE, annS]
```

Out[ ]=

$$\{y\, D_y - x\, D_x,\ (-2\, x + 4\, x^2\, y)\, D_x^2 + (-1 - 4\, x^2\, y^2)\, D_x + (y + 2\, x\, y^2)\}$$

In[ ]:=
```
ApplyOreOperator[ann, Exp[x y] + Sqrt[x y]]
```

Out[ ]=

$$\left\{y\left(e^{x\, y}\, x + \frac{x}{2\, \sqrt{x\, y}}\right) - x\left(e^{x\, y}\, y + \frac{y}{2\, \sqrt{x\, y}}\right),\right.$$
$$\left.(-2\, x + 4\, x^2\, y)\left(e^{x\, y}\, y^2 - \frac{y^2}{4\, (x\, y)^{3/2}}\right) + (-1 - 4\, x^2\, y^2)\left(e^{x\, y}\, y + \frac{y}{2\, \sqrt{x\, y}}\right) + (y + 2\, x\, y^2)\left(e^{x\, y} + \sqrt{x\, y}\right)\right\}$$

In[ ]:=
```
FullSimplify[%]
```

Out[ ]=

$$\{0,\ 0\}$$

In[ ]:=
```
ApplyOreOperator[ToOrePolynomial[LCLM, y[x]], f[1, x] + f[2, x]] // FullSimplify
```
Out[ ]=
```
0
```

In[ ]:=
```
? Annihilator
```
Out[ ]=

Symbol

Annihilator[expr, ops] computes annihilating relations for expr w.r.t. the given operator(s). It returns the Groebner basis of an annihilating ideal (with monomial order DegreeLexicographic). If expr is ∂–finite, the result will be a ∂–finite ideal. If expr is not recognized to be ∂–finite, there is still a chance to find at least some relations (in this case the ideal is not zero–dimensional which is indicated by a warning). Annihilator[expr] automatically determines for which operators relations exist. The relations are computed by executing the ∂–finite closure properties DFinitePlus, DFiniteTimes, and DFiniteSubstitute.

The expression expr can contain hypergeometric expressions, hyperexponential expressions, and algebraic expressions.

Additionally the following functions are recognized: AiryAi, AiryAiPrime, AiryBi, AiryBiPrime, AngerJ, AppellF1, ArcCos, ArcCosh, ArcCot, ArcCoth, ArcCsc, ArcCsch, ArcSec, ArcSech, ArcSin, ArcSinh, ArcTan, ArcTanh, ArithmeticGeometricMean, BellB, BernoulliB, BesselI, BesselJ, BesselK, BesselY, Beta, BetaRegularized, Binomial, CatalanNumber, ChebyshevT, ChebyshevU, Cos, Cosh, CoshIntegral, CosIntegral, EllipticE, EllipticF, EllipticK, EllipticPi, EllipticTheta, EllipticThetaPrime, Erf, Erfc, Erfi, EulerE, Exp, ExpIntegralE, ExpIntegralEi, Factorial, Factorial2, Fibonacci, FresnelC, FresnelS, Gamma, GammaRegularized, GegenbauerC, HankelH1, HankelH2, HarmonicNumber, HermiteH, Hypergeometric0F1, Hypergeometric0F1Regularized, Hypergeometric1F1, Hypergeometric1F1Regularized, Hypergeometric2F1, Hypergeometric2F1Regularized, HypergeometricPFQ, HypergeometricPFQRegularized, HypergeometricU, JacobiP, KelvinBei, KelvinBer, KelvinKei, KelvinKer, LaguerreL, LegendreP, LegendreQ, LerchPhi, Log, LogGamma, LucasL, Multinomial, NevilleThetaC, ParabolicCylinderD, Pochhammer, PolyGamma, PolyLog, qBinomial, QBinomial, qBrackets, qFactorial, QFactorial, qPochhammer, QPochhammer, Root, Sin, Sinc, Sinh, SinhIntegral, SinIntegral, SphericalBesselJ, SphericalBesselY, SphericalHankelH1, SphericalHankelH2, Sqrt, StirlingS1, StirlingS2, StruveH, StruveL, Subfactorial, WeberE, WhittakerM, WhittakerW, Zeta.

If expr contains the commands D and ApplyOreOperator then the closure property DFiniteOreAction is performed: Note the difference between

Annihilator[D[LegendreP[n, x], x], {S[n], Der[x]}] and expr = D[LegendreP[n, x], x]; Annihilator[expr, {S[n], Der[x]}].

Similarly, if expr contains Sum or Integrate then not Mathematica is asked to simplify the expression, but CreativeTelescoping is executed automatically on the summand (resp. integrand). For evaluating the delta part, Mathematica's FullSimplify is used; if it fails (or if you don't trust it), you can use the option Inhomogeneous –> True, in order to obtain an inhomogeneous recurrence (resp. differential equation).

In[ ]:=
```
Annihilator[StirlingS2[n, k], {S[n], S[k]}]
```

Annihilator: The expression StirlingS2[n, k] is not recognized to be ∂–finite. The result might not generate a zero–dimensional ideal.

Out[ ]=

$$\{S_n S_k + (-1 - k) S_k - 1\}$$

In[ ]:=
```
annB = Annihilator[Binomial[n, k], {S[n], S[k]}]
```

Out[ ]=

$$\{(1 + k) S_k + (k - n),\ (1 - k + n) S_n + (-1 - n)\}$$

In[ ]:=
```
deB = DFiniteRE2DE[%, {n, k}, {x, y}]
```

Out[ ]=

$$\{-x\,y\,D_x D_y + (y + y^2) D_y^2 - x\,D_x + (1 + 2\,y) D_y,\ (-x^2 + x^3) D_x^2 + (y + y^2) D_y^2 + (-2\,x + 3\,x^2) D_x + (1 + 2\,y) D_y + x,\ (-y^2 + x\,y^2 - y^3 + 2\,x\,y^3 + x\,y^4) D_y^3 + (-y + x\,y - 3\,y^2 + 7\,x\,y^2 + 6\,x\,y^3) D_y^2 + (-x + x^2) D_x + (1 - x + 2\,x\,y + 7\,x\,y^2) D_y + x\,y\}$$

In[ ]:=
```
Support[deB]
```

Out[ ]=

$$\{\{D_x D_y,\ D_y^2,\ D_x,\ D_y\},\ \{D_x^2,\ D_y^2,\ D_x,\ D_y,\ 1\},\ \{D_y^3,\ D_y^2,\ D_x,\ D_y,\ 1\}\}$$

In[ ]:=
```
Sum[Binomial[n, k] x^n y^k, {n, 0, Infinity}, {k, 0, Infinity}]
```

Out[ ]=

$$\frac{1}{1 - x - x\,y}$$

In[ ]:=

$$\text{deB1 = Annihilator}\left[\frac{1}{1 - x - x\,y}, \{\text{Der}[x], \text{Der}[y]\}\right]$$

Out[ ]=

$$\{(-1 + x + x\,y) D_y + x,\ (-1 + x + x\,y) D_x + (1 + y)\}$$

In[ ]:=
```
OreReduce[deB, deB1]
```

Out[ ]=

```
{0, 0, 0}
```

In[ ]:=
```
? LegendreP
```
Out[ ]=

Symbol

LegendreP[*n*, *x*] gives the Legendre polynomial $P_n(x)$.

LegendreP[*n*, *m*, *x*] gives the associated Legendre polynomial $P_n^m(x)$.

In[ ]:=
```
Annihilator[LegendreP[n, x], {S[n], Der[x]}]
```
Out[ ]=

$$\left\{(1+n)\,S_n+\left(1-x^2\right)D_x+(-x-n\,x),\ \left(-1+x^2\right)D_x^2+2\,x\,D_x+\left(-n-n^2\right)\right\}$$

In[ ]:=
```
Annihilator[LegendreP[n, x], {Der[x], S[n]}]
```
Out[ ]=

$$\left\{\left(1-x^2\right)D_x+(1+n)\,S_n+(-x-n\,x),\ (2+n)\,S_n^2+(-3\,x-2\,n\,x)\,S_n+(1+n)\right\}$$